MICROPROCESSORS
ENTC 4337

The following program listing, INTERR . ASM, illustrates an in​terrupt generated internally by the C31 timer 0.  The rate at which an interrupt occurs is determined without the use of the AIC. 

Program:


Source Code
;INTERR.ASM - DEMONSTRATES INTERNAL INTERRUPT WITHOUT THE AIC

         .start "intsect",0x809FC9 ;starting address for interrupt

         .start ".text",0x809A00   ;starting address for text

         .start ".data",0x809C00   ;starting address for data

         .sect  "intsect"     ;section for interrupt

         BR     ISR           ;interrupt vector TINT0

         .data                ;data section

PERIOD   .word  2000H         ;interrupt rate=12.5MHz/(2*PERIOD)

IE_REG   .word  100H          ;enable timer 0 (TINT0) for interrupt

PER_ADDR .word  808028H       ;(TLCK0) period register location

TCNTL    .word  2C1H          ;control register value

ST_REG   .word  2000H         ;set status register

OUTPUT   .word  0xA           ;initial output value

OUT_ADDR .word  0x809A30      ;output address

STACKS   .word  809F00h       ;init stack pointer 

         .entry BEGIN         ;start of code

         .text                ;assemble into text section

BEGIN    LDP    STACKS        ;init data page

         LDI    @STACKS,SP    ;SP -> 0809F00h 

         LDI    @PER_ADDR,AR0 ;TINT0 period register =>AR0

         LDI    @OUT_ADDR,AR1 ;output address =>AR1

         LDI    @PERIOD,R0    ;period value => R0

         STI    R0,*AR0--(8)  ;set TLCK0 period @ 808028H

         LDI    @TCNTL,R0     ;control register value =>R0

         STI    R0,*AR0       ;set TLCK0 global control @ 808020H

         LDI    @OUTPUT,R0    ;R0 = output value

         OR     @IE_REG,IE    ;enable TINT0 interrupt bit 8

WAIT     IDLE                 ;wait for interrupt

         BR     WAIT          ;branch to WAIT until interrupt

;  INTERRUPT VECTOR                                        

ISR      ADDI   2,R0          ;increment output value by 2

         STI    R0,*AR1++     ;store output value

         RETI                 ;return from interrupt

        .end                  ;end


Object Code

>>>> DSK3A C3x DSP Starter Kit Assembler Rev 1.18

>>>> (c) Copyright 1994-1995 Texas Instruments Incorporated

>>>> Wed Oct 04 17:08:28 2000

>>>> INTERR.ASM

>>>>

>>>>

>>>> PASS 1 Complete

>>>> Errors: 0  Warnings: 0

>>>>

0x00809802 nocode     ;INTERR.ASM - DEMONSTRATES INTERNAL INTERRUPT WITHOUT THE AIC

0x00809802 directive           .start "intsect",0x809FC9 ;starting address for interrupt

0x00809802 directive           .start ".text",0x809A00   ;starting address for text

0x00809802 directive           .start ".data",0x809C00   ;starting address for data

0x00809802 directive           .sect  "intsect"     ;section for interrupt

0x00809fc9 0x60809a0c          BR     ISR           ;interrupt vector TINT0

0x00809fca directive           .data                ;data section

0x00809c00 directive  PERIOD   .word  2000H         ;interrupt rate=12.5MHz/(2*PERIOD)

0x00809c00 0x00002000 <word>

0x00809c01 directive  IE_REG   .word  100H          ;enable timer 0 (TINT0) for interrupt

0x00809c01 0x00000100 <word>

0x00809c02 directive  PER_ADDR .word  808028H       ;(TLCK0) period register location

0x00809c02 0x00808028 <word>

0x00809c03 directive  TCNTL    .word  2C1H          ;control register value

0x00809c03 0x000002c1 <word>

0x00809c04 directive  ST_REG   .word  2000H         ;set status register

0x00809c04 0x00002000 <word>

0x00809c05 directive  OUTPUT   .word  0xA           ;initial output value

0x00809c05 0x0000000a <word>

0x00809c06 directive  OUT_ADDR .word  0x809A30      ;output address

0x00809c06 0x00809a30 <word>

0x00809c07 directive  STACKS   .word  809F00h       ;init stack pointer 

0x00809c07 0x00809f00 <word>

0x00809c08 directive           .entry BEGIN         ;start of code

0x00809c08 directive           .text                ;assemble into text section

0x00809a00 0x50700080 BEGIN    LDP    STACKS        ;init data page

0x00809a01 0x08349c07          LDI    @STACKS,SP    ;SP -> 0809F00h 

0x00809a02 0x08289c02          LDI    @PER_ADDR,AR0 ;TINT0 period register =>AR0

0x00809a03 0x08299c06          LDI    @OUT_ADDR,AR1 ;output address =>AR1

0x00809a04 0x08209c00          LDI    @PERIOD,R0    ;period value => R0

0x00809a05 0x15402808          STI    R0,*AR0--(8)  ;set TLCK0 period @ 808028H

0x00809a06 0x08209c03          LDI    @TCNTL,R0     ;control register value =>R0

0x00809a07 0x1540c000          STI    R0,*AR0       ;set TLCK0 global control @ 808020H

0x00809a08 0x08209c05          LDI    @OUTPUT,R0    ;R0 = output value

0x00809a09 0x10369c01          OR     @IE_REG,IE    ;enable TINT0 interrupt bit 8

0x00809a0a 0x06000000 WAIT     IDLE                 ;wait for interrupt

0x00809a0b 0x60809a0a          BR     WAIT          ;branch to WAIT until interrupt

0x00809a0c nocode     ;  INTERRUPT VECTOR                                        

0x00809a0c 0x02600002 ISR      ADDI   2,R0          ;increment output value by 2

0x00809a0d 0x15402101          STI    R0,*AR1++     ;store output value

0x00809a0e 0x78000000          RETI                 ;return from interrupt

0x00809a0f directive          .end                  ;end


Reference Table

>>>>

>>>> PASS 2 Complete

>>>> Errors: 0  Warnings: 0

>>>>

>>>> ENTRY 0x00809a00

>>>>

>>>> Symbol reference table           Type  Addressable

>>>> ref     Default_sect 0x00809802  1     1

>>>> ref          intsect 0x00809fc9  1     1

>>>> ref            .text 0x00809a00  1     1

>>>> ref            .data 0x00809c00  1     1

>>>> ref           PERIOD 0x00809c00  1     1

>>>> ref           IE_REG 0x00809c01  1     1

>>>> ref         PER_ADDR 0x00809c02  1     1

>>>> ref            TCNTL 0x00809c03  1     1

>>>> ref           ST_REG 0x00809c04  1     1

>>>> ref           OUTPUT 0x00809c05  1     1

>>>> ref         OUT_ADDR 0x00809c06  1     1

>>>> ref           STACKS 0x00809c07  1     1

>>>> ref            BEGIN 0x00809a00  1     1

>>>> ref             WAIT 0x00809a0a  1     1

>>>> ref              ISR 0x00809a0c  1     1

>>>>

>>>>        Output section start       end         length

>>>> sect     Default_sect 0x00809802  0x00809802  0x00000000

>>>> sect          intsect 0x00809fc9  0x00809fca  0x00000001

>>>> sect            .text 0x00809a00  0x00809a0f  0x0000000f

>>>> sect            .data 0x00809c00  0x00809c08  0x00000008

>>>>

>>>>

>>>> END DSK

Background:

1.  The interrupt rate is determined by a value set in the period register, or




rate =  12.5 MHz(2 x period)

2.  The period register at the memory address 808028 and the global regis​ter at the memory address 808020 are initialized. Bit 8 within the interrupt enable (IE) register TINT0 is enabled. Appendix A (Chassaing) contains information on these registers.

3.  Execution continues within a loop containing the two instructions WAIT IDLE and BR WAIT until an interrupt occurs. The counter register at memory address 808024 increments from 0,1,..., until it reaches the period value of 2000h set in the period register at which time an interrupt occurs. The counter register is reset to zero and is incremented again.

4.
On interrupt, execution proceeds to the interrupt service routine (ISR).  An initial value of 0xA 10 (decimal) set as output is incremented by two (within the interrupt service routine) and the result stored in memory location 809a30, the starting output address specified by OUT_ADDR.

5.
Execution returns to the WAIT loop until the next interrupt occurs.

Procedure:

1.
Run this program for one or two seconds, then stop/halt execution. Type memd 0x809a30 to verify the output values 12,14,16,18,....  The C31 should be reset first, before displaying the output values, if an old version of the DSK tools is used.  

Due to the interrupt structure, it is not possible to single-step and observe the counter register at 808024 incrementing, or to observe the sequence of the program counter (PC) illustrating the instruction to be executed next, specifically when the timer counter register equals the period register value.

2.   Type mm 0x809a30 100 0000h to clear the output memory contents.  This memory modifies the locations beginning at 0x809a30 with 0000h for 100 locations.

3.   Type reset to reset the system and reload the program.

4.   Now set a breakpoint at location 0x809a0a by typing sb 0x809a0a.  Run the program and notice that the program halts at 0x809a0a.  

5.   Check the contents of the following:




a.  stack pointer




b.  0x808028




c. 0x808020

6.   Type memd 0x809a30 and record the contents of 0x809a30.

7.   Rerun the program and note the contents of:




a.  AR1




b.  OUT_ADDR

8.   Repeat step 7 several times and comment upon your observations.

9.   Repeat steps 2 and 3 above.

10.  Now set a breakpoint at location 0x809a0c by typing sb 0x809a0c.  

11.  Run the program and notice that the program halts at 0x809a0c.  

12.  Check the value of the stack pointer.

Questions:

The following questions should be addressed in your lab report.

1. What are the I/O peripherals available on the TMS320C31?

2. Describe the interrupt process for the program you performed?

3. What is an interrupt, and what is the purpose of an interrupt?

4. What is meant by the term interrupt vector?

5. What is the purpose of setting breakpoint in your program?

6. How does the stack pointer change during the interrupt process?

7. What operation and register(s) are used to enable the interrupt in the program?

